
Beispiel für ein Exit-Programm vom Typ 2 (RPGLE)

     H COPYRIGHT(‘© menten GmbH, 2003‘)
     **-----------------------------------------------------------------------**
     **                                      tt                               **
     **           mmmmmm    eeeee   nnnnnn   tt     eeeee   nnnnnn            **
     **           m  m  m  e     e  n     n  tttt  e     e  n     n           **
     **           m  m  m  eeeeeee  n     n  tt    eeeeeee  n     n           **
     **           m  m  m  e        n     n  tt    e        n     n           **
     **           m  m  m   eeeeee  n     n  ttt    eeeeee  n     n           **
     **                                                                       **
     ** E D V  -  B e r a t u n g  G m b H                                    **
     **                                                                       **
     **                     Systemberatung                                    **
     **                  IBM System AS/400                                    **
     **           Individualprogrammierung                                    **
     **                                   menten GmbH                         **
     **                                   Hauptstrasse 136 - 140              **
     **                                   51465 Bergisch Gladbach             **
     **                                   Telefon  +49 2202 2399 0            **
     **                                   Telefax  +49 2202-2399 23           **
     **                                   Email    info@menten-gmbh.de        **
     **                                   Homepage http://www.menten-gmbh.de  **
     **-----------------------------------------------------------------------**
     ** PROGRAM NAME: EXITSNDEML                                              **
     **                                                                       **
     ** APPLICATION : CVTSPL command                                          **
     **                                                                       **
     ** LANGUAGE    : ILE RPG                                                 **
     **                                                                       **
     ** AUTHOR      : Ilja Kolosow                                            **
     **                                                                       **
     ** DATE WRITTEN: Sep. 2002                                               **
     **                                                                       **
     ** DESCRIPTION : Sample ‘TYPE 2‘ CVTSPL exit program for sending a       **
     **               stream file via e-mail using the SNDEMAIL command        **
     **               of i-effect                                             **
     **-----------------------------------------------------------------------**

      * Customer file

     FCustFile  IF   E           K DISK    RENAME(CUSTFILE:CUSTFILER)

     **-----------------------------------------------------------------------**
     ** PROTOTYPE FOR STANDARD EXIT PROGRAM PARAMETERS                        **
     **-----------------------------------------------------------------------**

     D Parameters      PR                  EXTPGM(‘EXITSNDEML‘)

      * Spooled file name

     D   iaSplFile                   10A

      * Name of the job which created the spooled file

     D   iaSplJob                    10A

      * User id of the job which created the spooled file

     D   iaSplUser                   10A
      * Job number of the job which created the spooled file

     D   iaSplJobNo                   6A

      * Spooled file number of the spooled file

     D   iiSplNbr                    10I 0



      * Spooled file user data

     D   iaUsrDta                    10A

      * Name of the stream file created

     D   iaStmFile                  128A

      * IFS directory in which stream file was created

     D   iaDir                      256A

      * Page number of the first page in the range processed

     D   iiFirstPage                 10I 0

      * Page number of the last page in the range processed

     D   iiLastPage                  10I 0

      * Exit program parameter count

     D   iiParmCount                 10I 0

      * User-definable exit program parameter strings. Up to 150 of these
      * may follow. Define as many as you need.

     D   iaUserParam1               999A         OPTIONS(*NOPASS)
     D                                           VARYING

     D   iaUserParam2               999A         OPTIONS(*NOPASS)
     D                                           VARYING

     D   iaUserParam3               999A         OPTIONS(*NOPASS)
     D                                           VARYING

     **-----------------------------------------------------------------------**
     ** PROTOTYPE FOR QCMDEXC                                                 **
     **-----------------------------------------------------------------------**

     D Command         PR                  ExtPgm(‘QCMDEXC‘)
     D   iaCommand                32767A   CONST OPTIONS(*VARSIZE)
     D   inCmdLen                    15P 5 CONST

     **-----------------------------------------------------------------------**
     ** STANDARD EXIT PROGRAM PARAMETERS                                      **
     **-----------------------------------------------------------------------**

     D Parameters      PI
     D   iaSplFile                   10A
     D   iaSplJob                    10A
     D   iaSplUser                   10A
     D   iaSplJobNo                   6A
     D   iiSplNbr                    10I 0
     D   iaUsrDta                    10A
     D   iaStmFile                  128A
     D   iaDir                      256A
     D   iiFirstPage                 10I 0
     D   iiLastPage                  10I 0
     D   iiParmCount                 10I 0
     D   iaUserParam1               999A         OPTIONS(*NOPASS)
     D                                           VARYING
     D   iaUserParam2               999A         OPTIONS(*NOPASS)
     D                                           VARYING
     D   iaUserParam3               999A         OPTIONS(*NOPASS)
     D                                           VARYING

     D waCustomerNo    S             10A
     D waCommand       S           1024A



     D wnCmdLen        S             15P 5

     D QUOTE           C                   ‘‘‘‘

      * This example assumes that the EXITPGMPRM, EXITPGMPOS and EXITPGMKEY
      * parameters have been used to extract the customer number from the
      * spooled file, and that the first user-definable parameter contains the
      * customer number (assumed to be 10 characters for the purposes of
      * this example).

      * Make sure at least one parameter was passed and that the parameter
      * passed is not to short

     C                   IF           iiParmCount < 1
     C                             or %len(iaUserParam1) < %size(waCustomerNo)
      * Insert appropriate error handling here
     C                   RETURN
     C                   ENDIF

      * Extract the customer number from the parameter string

     C                   EVAL      waCustomerNo = %subst(iaUserParam1      :
     C                                                   1                 :
     C                                                   %len(waCustomerNo))

      * Retrieve the customer‘s e-mail address from the customer file

     C     waCustomerNo  CHAIN     CustFile
     C                   IF        not %found(CustFile)
      * Insert appropriate error handling here
     C                   RETURN
     C                   ENDIF

      * Build the SNDEMAIL command
     C                   EVAL      waCommand = ‘SNDEMAIL ‘
     C                                       + ‘RECIPIENT(‘ + QUOTE
      * „CustEMail“ is assumed to be the customer‘s email address field
      * from the customer file
     C                                       + %trim(CustEMail) + QUOTE
     C                                       + ‘) ‘

      * DSTD parameter below is the „subject“ line that will appear for
      * the e-mail. Substitute your choice of subject line
     C                                       + ‘SUBJECT(‘
     C                                       + QUOTE
     C                                       + ‘Your invoice from i-effect‘
     C                                       + QUOTE
     C                                       + ‘) ‘
      * MSG parameter below is a brief message to accompany the attachment.
      * Substitute your choice of message.
     C                                       + ‘MSGTEXT(‘
     C                                       + QUOTE
     C                                       + ‘Your invoice is‘
     C                                       + ‘ attached. Thank you for your ‘
     C                                       + ‘custom.‘
     C                                       + QUOTE
     C                                       + ‘) ‘
      * The document name and folder names are passed into this program from
      * CVTSPL. The document name if generated from the OUTPFILE parameter
      * with the addition of a numeric suffix for each spooled file created.
     C                                       + ‘ATTACH(‘ + QUOTE
     C                                       + %trim(%trim(iaDir) + ‘/‘
     C                                       + %trim(iaStmFile)) + QUOTE
     C                                       + ‘)‘
     C                   eval      wnCmdLen = %len(waCommand)

      * e-mail the stream file to the customer
     C                   CALLP(E)  Command(waCommand:wnCmdLen)

     C                   IF        %error



      * Insert appropriate error handling here
     C                   ELSE
      * Send OK message
     C                   ENDIF
     C                   RETURN


